

1. BASICS

We quickly recall some of the definitions from the first lecture. Let $R = k[x_1, \dots, x_n]$ be a polynomial ring over k . A *monomial* has the form

$$x^\alpha = x_1^{a_1} x_2^{a_2} \cdots x_n^{a_n} \text{ with } a_i \in \mathbb{N}.$$

A monomial is *squarefree* if each $a_i = 0$ or 1 . An ideal is a (*squarefree*) *monomial ideal* if I is generated by (squarefree) monomial ideals.

Example 1.1. Let $R = k[x_1, x_2, x_3, x_4]$. The ideal $I = (x_1^6, x_2^3 x_4, x_4)$ is a monomial ideal, while $I = (x_1 x_2, x_2 x_3, x_3 x_4, x_4 x_1)$ is a squarefree monomial ideal.

Lemma 1.2. *Every monomial ideal is generated by a unique minimal set of generators, and this set is finite.*

For a proof, see Lemma 1.2 of [MS].

Question 1.3. *How do I enter a monomial ideal into Macaulay 2?*

Answer. As in most operations, you must first specify a polynomial ring which contains the ideal I . To enter the ideal, use the `monomialIdeal` command. For example:

```
i1 : R = QQ[x_1..x_5]
o1 = R
o1 : PolynomialRing
i2 : I = monomialIdeal(x_1*x_2,x_2*x_3*x_4)
o2 = monomialIdeal (x *x , x *x *x )
      1 2 2 3 4
o2 : MonomialIdeal of R
```

You could also use the command `ideal`. However, if you will only be using monomial ideals, it is better to use the command `monomialIdeal` since some calculations are better optimized for monomial ideals. For example, compare the two inputs:

```
i3 : J1= monomialIdeal(x_1*x_2,x_2*x_3*x_4,x_2)
o3 = monomialIdeal(x )
      2
o3 : MonomialIdeal of R
i4 : J2 =ideal(x_1*x_2,x_2*x_3*x_4,x_2)
o4 = ideal (x *x , x *x *x , x )
      1 2 2 3 4 2
```

o4 : Ideal of R

Notice from the output that the `monomialIdeal` command recognizes the minimal generators of the ideal, while the `ideal` command does not find a list of the minimal generators.

2. SIMPLICIAL COMPLEXES AND STANLEY-REISNER RINGS

During the first lecture, you were introduced to simplicial complexes. We now describe how to use *Macaulay 2* to study and describe simplicial complexes.

Definition 2.1. An (*abstract*) *simplicial complex* Δ on $V = \{x_1, \dots, x_n\}$ is a subset of the power set of V such that


- $\{x_i\} \in \Delta$ for each i ;
- if $F \in \Delta$ and $G \subseteq F$, then $G \in \Delta$.

Remark. If $\Delta \neq \emptyset$, then $\emptyset \in \Delta$. We usually omit \emptyset when writing the elements of Δ .

Example 2.2. Let $V = \{x_1, x_2, x_3, x_4, x_5\}$. The follow set Δ is an example of a simplicial complex:

$$\Delta = \{\{x_1, x_2, x_3\}, \{x_1, x_2, x_4\}, \{x_1, x_3, x_4\}, \{x_2, x_3, x_4\}, \{x_1, x_2\}, \{x_1, x_3\}, \{x_1, x_4\}, \\ \{x_2, x_3\}, \{x_2, x_4\}, \{x_2, x_5\}, \{x_3, x_4\}, \{x_3, x_5\}, \{x_1\}, \{x_2\}, \{x_3\}, \{x_4\}, \{x_5\}\}.$$

We can visualize Δ has a hollow tetrahedron with an edge attached at both x_2 and x_3 . Both of these edges are also attached to the vertex x_5 . Here is a picture:



Definition 2.3. The elements of Δ are called *faces*. The maximal faces under inclusion are called *facets*.

We use the notation $|S|$ to denote the number of elements in a set S .

Definition 2.4. If $F \in \Delta$ is a face, then the dimension of F is $\dim F = |F| - 1$. If $F = \emptyset$, we set $\dim F = -1$. The *dimension* of Δ is $\dim \Delta = \max\{\dim F \mid F \in \Delta\}$.

Definition 2.5. If Δ is a simplicial complex, then set

$$f_i = f_i(\Delta) = \text{number of faces of dimension } i.$$

If $\dim \Delta = d$, then the f -vector of Δ is the d -tuple $f(\Delta) = (f_0, f_1, \dots, f_{d-1})$.

Remark. $f_{-1} = 1$ since $\emptyset \in \Delta$ and $\dim \emptyset = -1$. Also $f_0 = |V|$, the number of vertices.

Definition 2.6. A simplicial complex Δ is *pure* if all its facets have the same dimension.

Example 2.7. The f -vector of the simplicial complex of Example 2.2 is $f(\Delta) = (5, 8, 4)$. Since Δ has 4 facets of dimension 2, and 2 facets of dimension 1, Δ is not a pure simplicial complex.

Now that we have introduced some terminology, let's describe how to do some of these things using *Macaulay 2*.

Question 2.8. *How do I input a simplicial complex into Macaulay 2?*

Answer. *Macaulay 2* contains a package specifically written for dealing with simplicial complexes. To use this package, you first need to load the package using the following command:

```
i5 : loadPackage "SimplicialComplexes";
i6 : R = ZZ[x_1..x_5] -- introduce polynomial ring
o6 = R
o6 : PolynomialRing
```

Once you have loaded the package, you need to define a polynomial ring where the indeterminates of the ring correspond to the vertex set of Δ . In the above, we have already defined the polynomial ring with five variables.

To enter the simplicial complex, enter the facets as list. When entering a facet, write the facet as a squarefree monomial. E.g., the facet $\{a, b, c\}$ is inputted as $a*b*c$. Then apply the command `simplicialComplex` to the list.

```
i7 : L = {x_1*x_2*x_3, x_1*x_2*x_4, x_1*x_3*x_4, x_2*x_3*x_4, x_2*x_5, x_3*x_5}
      -- write each facet as a monomial
o7 = {x *x *x , x *x *x , x *x *x , x *x *x , x *x , x *x }
      1 2 3 1 2 4 1 3 4 2 3 4 2 5 3 5
o7 : List
i8 : Delta = simplicialComplex L
o8 = | x_3x_5 x_2x_5 x_2x_3x_4 x_1x_3x_4 x_1x_2x_4 x_1x_2x_3 |
o8 : SimplicialComplex
```

Note that `Delta` is the simplicial complex of Example 2.2.

Question 2.9. *I've inputted my simplicial complex Δ ; how do I now find $\dim \Delta$? the f -vector? the facets of Δ ?*

Answer. *Macaulay 2* contains functions for each of the above desired operations (remember, you must first load the `simplicialComplexes` package). For dimension, simply input

```
i9 : dim Delta
```

```
o9 = 2
```

For the f -vector, you can use the command `fVector`:

```
i10 : fVector Delta
```

```
o10 = HashTable{-1 => 1}
      0 => 5
      1 => 8
      2 => 4
```

```
o10 : HashTable
```

The output is listed a table; so, for our simplicial complex, we have five 0-dimensional faces, eight 1-dimensional faces, and four 2-dimensional faces. (This agrees with Example 2.7.) The command `facets` will output a matrix whose entries consist of the facets of the simplicial complex.

```
i11 : facets Delta -- all facets of Delta
```

```
o11 = | x_3x_5 x_2x_5 x_2x_3x_4 x_1x_3x_4 x_1x_2x_4 x_1x_2x_3 |
```

```
      1      6
o11 : Matrix R <--- R
```

Macaulay 2 actually allows you to list all the faces of Δ of any given dimension with the `faces` command. Note that the output is a matrix.

```
i12 : faces(0,Delta) -- all 0-dim faces
```

```
o12 = | x_1 x_2 x_3 x_4 x_5 |
```

```
      1      5
o12 : Matrix R <--- R
```

```
i13 : faces(1,Delta) -- all 1-dim faces
```

```
o13 = | x_1x_2 x_1x_3 x_1x_4 x_2x_3 x_2x_4 x_2x_5 x_3x_4 x_3x_5 |
```

```
      1      8
o13 : Matrix R <--- R
```

```
i14 : faces(4,Delta) -- all 4-dim faces (there are none!)
```

```
o14 = 0
```

```
      1
o14 : Matrix R <--- 0
```

The package `simplicialComplexes` also contains a command to check if Δ is pure.

```
i15 : isPure(Delta)
```

```
o15 = false
```

Again, this answer agrees with our earlier observation that Δ is not pure.

Let's return to f -vectors. Given a d tuple $f = (f_0, f_1, \dots, f_{d-1}) \in \mathbb{N}^d$, it is natural to ask if there exists a simplicial complex Δ with f -vector $f(\Delta) = f$. Kruskal and Katona independently gave a characterization of what vectors could be the f -vector of a simplicial complex. To state their result, we need the following result. In the sequel we set $\binom{l}{k} = 0$ if $0 \leq l < k$.

Lemma 2.10. *Let d be a positive integer. Any $a \in \mathbb{N}$ can be written uniquely in the form*

$$a = \binom{k(d)}{d} + \binom{k(d-1)}{d-1} + \dots + \binom{k(j)}{j}$$

where $k(d) > k(d-1) > \dots > k(j) \geq j \geq 1$. We refer to the above sum as the d -th Macaulay representation of a .

See Lemma 4.2.6 of [BH] for a proof. We now define the function ${}^{(d)} : \mathbb{N} \rightarrow \mathbb{N}$ as follows: for any $a \in \mathbb{N}$, set

$$a^{(d)} = \binom{k(d)}{d+1} + \binom{k(d-1)}{d} + \dots + \binom{k(j)}{j+1}$$

where $k(d), \dots, k(j)$ are the numbers that appear in the d th Macaulay representation of a .

Example 2.11. Let $a = 23$ and $d = 3$. The 3rd Macaulay representation of 23 is $23 = \binom{6}{3} + \binom{3}{2} = 20 + 3$. Thus $23^{(3)} = \binom{6}{4} + \binom{3}{3} = 15 + 1 = 16$.

Theorem 2.12. *A tuple $f = (f_0, \dots, f_{d-1}) \in \mathbb{N}^d$ is the f -vector of some $(d-1)$ dimension simplicial complex if and only if*

$$0 < f_{i+1} \leq f_i^{(i+1)} \text{ for } 0 \leq i \leq d-2.$$

This theorem is Theorem 2.1 of [S].

Question 2.13. *I want to test if a tuple is a f -vector. How do I do this?*

Answer. For this question, it is probably best to break this problem into two tasks. The first task is to write a program that receives a and d as input and returns $a^{(d)}$. The second task is to check if a given tuple satisfies the conditions of Theorem 2.12. Here is a program that does the first task:

```
i16 : aUpperd = (a,d)-> (
      k = {};
      for i from 0 to (d-1) do (
        ki = 0;
        while (a >= binomial(ki,d-i)) do ki = ki+1;
        k = append(k,ki-1);
        a = a - binomial(ki-1,d-i);
      );
      l = sum toList apply(0..(d-1),j->binomial(k_j,d+1-j));
      return(l)
    );
```

```
i17 : aUpperd(23,3)
```

```
o17 = 16
```

We have checked our program by calculating $23^{(3)}$. The output is 16, as we expected. This routine is the basis for our second routine that checks whether a tuple of \mathbb{N}^d is a valid f -vector.

```
i18 : validfVector=f->(
      for j from 0 to #f-2 do (
        if (f_(j+1) > aUpperd(f_j,j+1)) then return (false);
      );
      return(true);
    );
```

```
i19 : validfVector(5,8,4)
```

```
o19 = true
```

```
i20 : validfVector(6,16,18)
```

```
o20 = false
```

We know that $f = (5, 8, 4)$ is a valid f -vector since $f = f(\Delta)$, where Δ is the simplicial complex of Example 2.2. Also, to see that $f = (6, 16, 18)$ cannot be a f -vector, notice that if there was such a simplicial complex, it would have to have 6 vertices. But between 6 vertices we can have at most $\binom{6}{2} = 15$ edges, so $f_1 \leq 15$. However, our f has $f_1 = 16$.

As mentioned in the first lecture, we can associate to any simplicial complex Δ on $V = \{x_1, \dots, x_n\}$ the following squarefree monomial ideal:

$$I_\Delta = (x_{i_1} \cdots x_{i_r} \mid \{x_{i_1}, \dots, x_{i_r}\} \notin \Delta) \subseteq R = k[x_1, \dots, x_n].$$

The ideal I_Δ is generated by the minimal non-faces of Δ . The ideal I_Δ is called the *Stanley-Reisner ideal*, and R/I_Δ is the *Stanley-Reisner ring*.

Conversely, one can associate to a squarefree monomial ideal I a simplicial complex Δ by

$$\Delta = \{\{x_{i_1}, \dots, x_{i_r}\} \mid x_{i_1} \cdots x_{i_r} \text{ is a squarefree monomial not in } I\}.$$

As a quick exercise, you should verify that Δ is indeed a simplicial complex.

In the `simplicialComplexes` package, *Macaulay 2* contains functions for finding I_Δ given Δ , and Δ given I . We describe how this works

Question 2.14. *I have a simplicial complex Δ . How do I find I_Δ with Macaulay 2?*

Answer. Suppose you have entered your simplicial complex into *Macaulay 2* using the methods described above. The command `monomialIdeal` applied to the simplicial complex will then return the ideal I_Δ . For example, if we use `Delta` to represent the simplicial complex of Example 2.2 we can find the associated ideal I_Δ as follows:

```
i21 : monomialIdeal Delta
```

```
o21 = monomialIdeal (x *x *x *x , x *x , x *x *x , x *x )
      1 2 3 4   1 5   2 3 5   4 5
```

```
o21 : MonomialIdeal of R
```

The output is a monomial ideal. Note that $x_1x_2x_3x_4 \in I_\Delta$ since this monomial corresponds to the face $\{x_1, x_2, x_3, x_4\}$; in our example, the tetrahedron formed by x_1, x_2, x_3, x_4 is hollow,

so Δ is missing the face $\{x_1, x_2, x_3, x_4\}$. As well, there is no edge between x_1 and x_5 , so this is why x_1x_5 appears as a generator.

You could also use the command `ideal Delta` but then output is treated as an ideal and not a monomial ideal by *Macaulay 2*.

Question 2.15. *I have a squarefree monomial ideal. How do I find its associated simplicial complex Δ ?*

Answer. Enter the generators of the squarefree monomial ideal using the `monomialIdeal` command. Then apply the command `simplicialComplex`:

```
i22 : R = QQ[x_1..x_6]
o22 = R
o22 : PolynomialRing
i23 : I = monomialIdeal (x_1*x_2,x_2*x_3,x_3*x_4,x_4*x_5,x_5*x_6,x_6*x_1)
o23 = monomialIdeal (x_1*x_2, x_2*x_3, x_3*x_4, x_4*x_5, x_5*x_6, x_6*x_1)
o23 : MonomialIdeal of R
i24 : D = simplicialComplex I
o24 = | x_2x_4x_6 x_3x_6 x_1x_3x_5 x_2x_5 x_1x_4 |
o24 : SimplicialComplex
```

You can now employ any of the commands we mentioned earlier. For example, we can find the associated f -vector:

```
i25 : fVector D
o25 = HashTable{-1 => 1}
      0 => 6
      1 => 9
      2 => 2
o25 : HashTable
```

As seen in the first lecture, the Hilbert series of R/I_Δ contains information about Δ . We consider another similar result that relates the dimension of Δ to the dimension of R/I_Δ . See Theorem 5.1.4 of [BH] for a proof.

Theorem 2.16. *Let $R = k[x_1, \dots, x_n]$ with k a field, and Δ a simplicial complex on $V = \{x_1, \dots, x_n\}$. Then*

$$\dim R/I_\Delta = \dim \Delta + 1.$$

Note the two uses of dimension in the statement of the theorem. The dimension on the left hand side is the Krull dimension. On the right, dimension is used to denote the dimension of a simplicial complex.

Question 2.17. *How do I find the dimension of a ring in Macaulay 2?*

Answer. The appropriate command is `dim`. For example, using I and R as in the previous question we get

```
i26 : S = R/I
o26 = S
o26 : QuotientRing
i27 : dim S -- dimension of ring
o27 = 3
i28 : dim D -- dimension of simplicial complex
o28 = 2
```

In our example, we see that the dimension of R/I_Δ is indeed one more than the dimension of Δ . Note that the command `dim I` will also return the dimension of R/I .

As a warning, to make use of Theorem 2.16 you must ensure that the coefficient ring of R is a field. For example, if we repeat the above example, but replace the coefficients with \mathbb{Z} , then the identity of Theorem 2.16 no longer holds.

```
i29 : T = ZZ[x_1..x_6] -- incorrect example
o29 = T
o29 : PolynomialRing
i30 : I = monomialIdeal (x_1*x_2,x_2*x_3,x_3*x_4,x_4*x_5,x_5*x_6,x_6*x_1)
o30 = monomialIdeal (x *x , x *x , x *x , x *x , x *x , x *x )
           1 2   2 3   3 4   4 5   1 6   5 6
o30 : MonomialIdeal of T
i31 : D = simplicialComplex I
o31 = | x_2x_4x_6 x_3x_6 x_1x_3x_5 x_2x_5 x_1x_4 |
o31 : SimplicialComplex
i32 : B = T/I
o32 = B
o32 : QuotientRing
i33 : dim B
o33 = 4
i34 : dim D -- dim B != dim D + 1
o34 = 2
```


3. RESOLUTIONS AND BETTI NUMBERS

Let $R = k[x_1, \dots, x_n]$ be a polynomial ring over an arbitrary field k . If I is a homogeneous ideal of R , then associated to R/I is a *minimal graded free resolution* of the form

$$0 \rightarrow \bigoplus_j R(-j)^{\beta_{l,j}(R/I)} \rightarrow \bigoplus_j R(-j)^{\beta_{l-1,j}(R/I)} \rightarrow \dots \rightarrow \bigoplus_j R(-j)^{\beta_{1,j}(R/I)} \rightarrow R \rightarrow R/I \rightarrow 0$$

where the maps are exact, $l \leq n$, and $R(-j)$ is the R -module obtained by shifting the degrees of R by j . The number $\beta_{i,j}(R/I)$, the i, j th *graded Betti number* of R/I , is an invariant of R/I that equals the number of minimal generators of degree j in the i th syzygy module of R/I . The i th *Betti number* is $\beta_i(R/I) = \sum_{j \in \mathbb{N}} \beta_{i,j}(R/I)$.

One of the important questions in commutative algebra is to describe the graded Betti numbers of a monomial ideal. Ideally, we would like a formula for all the numbers that depends upon only knowing the generators of I . (Unfortunately, this is too much to expect since some of the numbers will depend upon the characteristic of the field k .)

In tomorrow's lecture, you will hear more about resolutions. Here, we quickly introduce some of the basic commands in *Macaulay 2* dealing with resolutions.

Question 3.1. *I have a homogeneous ideal; how do I find its resolution?*

Answer. For an example, consider the defining ideal of the following five points in \mathbb{P}^2 :

$$P_1 = [1 : 0 : 0], P_2 = [0 : 1 : 0], P_3 = [0 : 0 : 1], P_4 = [1 : 1 : 1], P_5 = [19 : 64 : 42].$$

If $R = k[x, y, z]$ is the coordinate ring of \mathbb{P}^2 , then the defining ideal of $P = [a : b : c]$ is $I_P = (ay - bx, az - cx)$. So, if $X = \{P_1, \dots, P_5\}$, then the defining ideal of X is $I_X = \bigcap_{i=1}^5 I_{P_i}$. We enter this ideal as follows.

```
i35 : R = QQ[x,y,z] -- coordinate ring of P^2
o35 = R
o35 : PolynomialRing

i36 : IX = intersect(ideal(y,z),ideal(x,z),ideal(x,y),ideal(y-x,z-x),ideal(19*y-64*x,19*z-42*x))
o36 = ideal (945x*y - 736x*z - 209y*z, 945y z + 704x*z - 1649y*z , 1890x z - 2327x*z + 437y*z )
o36 : Ideal of R

The command res I will then return the non-graded resolution, i.e., it will return the ranks
(i.e., the numbers  $\beta_i(R/I_X)$ ) of the free modules in the minimal free resolution of  $R/I_X$ .

i37 : G = res IX
o37 = R <-- R <-- R <-- 0
      1      3      2
      0      1      2      3
o37 : ChainComplex
```

So, $\beta_0(R/I_X) = 1$, $\beta_1(R/I_X) = 3$, and $\beta_2(R/I_X) = 2$. Note that the output truncates the beginning of the resolution, that is, it omits the “ $0 \leftarrow R/I_X \leftarrow$ ”. If you want to know the maps in the resolution, they can be recovered as follows:

```
i38 : G.dd
```

```

      1                                     3
o38 = 0 : R <----- R : 1
      | xy-736/945xz-209/945yz x2z-2327/1890xz2+437/1890yz2 y2z+704/945xz2-1649/945yz2 |

      3                                     2
1 : R <----- R : 2
      {2} | -xz+1909/1890z2 -yz+913/945z2 |
      {3} | y-736/945z      -704/945z      |
      {3} | -437/1890z      x-209/945z      |

      2
2 : R <----- 0 : 3
      0

```

```
o38 : ChainComplexMap
```

The maps are outputted as matrices.

Question 3.2. *I want the graded Betti numbers of my ideal I ; how do I use Macaulay 2 to find these numbers?*

Answer. The Betti numbers of an ideal can be extracted with the command `betti`. We continue to use I_X as in the last example:

```
i39 : betti res IX
```

```

o39 = total: 1 3 2
      0: 1 . .
      1: . 1 .
      2: . 2 2

```

The output of `betti res I` is sometimes called the *Betti diagram of I* . The top row lists the total number of generators in each syzygy module (compare with the ranks of the free modules that appear with the command `res I`). The entry in column i and row $j - i$ (the counting of each row and each column starts at 0 instead of 1) corresponds to the number $\beta_{i,j}(R/I)$. For example, in the second column and second row there is a 2. So $\beta_{2,2+2}(R/I_X) = \beta_{2,4}(R/I_X) = 2$.

Question 3.3. *I just want to know a specific Betti number $\beta_{i,j}(R/I)$. How do I do this?*

Answer. The output of `betti res I` is not stored as a matrix, so you cannot just ask for the entry in the matrix at position $(i, j - i)$. The following script, however, will return $\beta_{i,j}(R/I)$.

```

i40 : BettiIJ = (i,j,I)->(
      G = res I;
      B = G_i;
      B = flatten degrees B;
      t = tally B;
      if (t #? j) then return (t#j)
      else return(0);
    );

```

```
i41 : BettiIJ(2,3,IX)
```

```
o41 = 0
```

```
i42 : BettiIJ(2,4,IX)
```

```
o42 = 2
```

Two important invariants associated with a graded resolution are the regularity and the projective dimension.

Definition 3.4. Let I be a homogeneous ideal of $R = k[x_1, \dots, x_n]$. The *regularity* of R/I , denoted $\text{reg}(R/I)$, is the number

$$\text{reg}(R/I) = \max\{j - i \mid \beta_{i,j}(R/I) \neq 0\}.$$

The *projective dimension* of R/I , denoted $\text{pdim}(R/I)$, is the number

$$\text{pdim}(R/I) = \max\{i \mid \beta_{i,j}(R/I) \neq 0 \text{ for some } j\}.$$

The regularity measures the “width” of the resolution, while the projective dimension measures the “length” of the resolution. These invariants have a natural interpretation in terms of the Betti diagram. The regularity corresponds to the number of rows in the Betti diagram (counting rows starting with 0), while the projective dimension is the number of columns (again, start your counting with 0).

Question 3.5. *I have a homogeneous ideal I , and I want to know $\text{reg}(R/I)$ and $\text{pdim}(R/I)$. Are there commands to find the regularity and projective dimension?*

Answer. As a first step, you first need to input R/I so that *Macaulay 2* views R/I as an R -module, not simply a quotient ring. Using the ideal $I = I_X$ as above, we turn R/I into an R -module as follows:

```
i43 : B = coker gens IX
```

```
o43 = cokernel | 945xy-736xz-209yz 945y2z+704xz2-1649yz2 1890x2z-2327xz2+437yz2 |
```

```
o43 : R-module, quotient of R
```

In the above $B = R/I$, but it is stored as an R -module. We can now find the regularity and projective dimension of R/I as follows:

```
i44 : regularity(B)
```

```
o44 = 2
```

```
i45 : pdim(B)
```

```
o45 = 2
```

Note that this agrees with Betti diagram of R/I which had two rows and two columns (counting from 0 instead of 1).

We end our quick discussion of resolution by introducing a class of ideals with a special type of resolution. These ideals are related to today’s tutorials.

Definition 3.6. Let I be a homogeneous ideal of R , and suppose that all of the minimal generators of I have the same degree, say d . We say that I has a *linear resolution* if for all $i \geq 1$, $\beta_{i,j}(R/I) = 0$ if $j \neq i + d - 1$.

Note that if I has a linear resolution, and $\beta_{i,j}(R/I)$ is a nonzero graded Betti number, then $j = i + d - 1$. Using the Betti diagram, it is easy to recognize if an ideal has a linear resolution: all the nonzero entries of the Betti diagram (except for the 1 in the $(0,0)$ th spot) appear on the $(d - 1)$ th row. The ideal I_X does not have a linear resolution since it has nonzero entries on the first and second row.

Here is an example of an ideal that has a linear resolution.

```
i46 : R = QQ[x,y,z,w]
o46 = R
i47 : I = (ideal(x,y,z,w))^4;
o47 : Ideal of R
i48 : betti res I
o48 = total: 1 35 84 70 20
      0: 1 . . . .
      1: . . . . .
      2: . . . . .
      3: . 35 84 70 20
```

Question 3.7. *How do I check if an ideal has a linear resolution?*

Answer. Suppose that $\{F_1, \dots, F_s\}$ is a minimal set of generators for I . Let $d = \min\{\deg F_i\}_{i=1}^s$. It is not difficult to prove that I has a linear resolution if and only if $\text{reg}(R/I) = d - 1$. Using this fact, the following script allows us to test if an ideal has a linear resolution.

```
i49 : verifyLinearRes=I->(
      D = min flatten degrees I;
      B = coker gens I;
      R = regularity(B);
      if D-1 != R then return (false);
      return(true);
    );
i50 : verifyLinearRes(I)
o50 = true
i51 : verifyLinearRes(IX)
o51 = false
```

4. GENERIC INITIAL IDEALS

Let $R = k[x_1, \dots, x_n]$ be a polynomial ring, and let $>$ denote a term order. Let $x^\alpha = x_1^{a_1} x_2^{a_2} \cdots x_n^{a_n}$ denote a monomial of R , and let $F = c_{\alpha_0} x^{\alpha_0} + \cdots + c_{\alpha_s} x^{\alpha_s}$ be a polynomial

of R . Suppose that x^α is the largest monomial that appears in F under the term order $>$. Then the *initial term* of a F with respect to $>$ is defined to be $\text{in}_>(F) = c_\alpha x^\alpha$. (or $\text{in}(F)$ if $>$ is clear). The *initial ideal* of an ideal I is then $\text{in}_>(I) = (\text{in}(f) \mid f \in I)$. Again, if $>$ is clear, we omit simply write $\text{in}(I)$.

As noted in the first lecture, if $I = (f_1, \dots, f_r)$ and $\text{in}(I) = (\text{in}(f_1), \dots, \text{in}(f_r))$, then f_1, \dots, f_r is called a *Gröbner Basis* of I .

Let $GL_n(k)$ denote the general linear group of order n over k , i.e., all the $n \times n$ invertible matrices with entries in k . Given a matrix $g = (g_{ij}) \in GL_n(k)$, and a polynomial $F = F(x_1, \dots, x_n) \in R$, we can let g act on F as follows:

$$g \cdot F = F(gx_1, \dots, gx_n) \text{ where } gx_i = \sum_{j=1}^n g_{ij}x_j.$$

Now fix an ideal I of R . Every matrix $g \in GL_n(k)$ results in an initial ideal $\text{in}(g \cdot I)$ where by $g \cdot I = (g \cdot f \mid f \in I)$. We say that two matrices g and g' are equivalent if $\text{in}(g \cdot I) = \text{in}(g' \cdot I)$. This is an equivalence relation on $GL_n(k)$, and thus the equivalence classes partition the group $GL_n(k)$. One of these partitions is quite “large” in the following sense.

Lemma 4.1. *For a fixed I and term order $>$, one of the equivalence classes is a nonempty Zariski open subset U inside $GL_n(k)$*

For each $g \in U$, the Zariski open subset, the initial ideal $\text{in}(g \cdot I)$ is the same ideal.

Definition 4.2. Fix a term order $>$ on R , let I be an ideal of R , and let g be any element of the open Zariski subset of Lemma 4.1. The initial ideal $\text{in}_>(g \cdot I)$ is called the *generic initial ideal* of I for the term order $>$. It is denoted $\text{gin}_>(I) = \text{in}_>(g \cdot I)$.

Roughly speaking, the generic initial ideal is the ideal we should expect if we pick a “random” matrix $g \in GL_n(k)$ and form $\text{in}(g \cdot I)$.

Question 4.3. *How do I find the generic initial ideal with Macaulay 2?*

Answer. There is no *Macaulay 2* command to perform this operation. However, you can use the following “quick-and-dirty method” for getting the generic initial ideal:

```
--gin code from a tutorial by
--Mark Green and Mike Stillman
i52 : gin = method();

i53 : gin Ideal := I -> (
    S:=ring I;
    StoS:=map(S,S,random(S^{0},S^{numgens S:-1}));
    monomialIdeal StoS I);

i54 : gin MonomialIdeal := I -> gin ideal I;

i55 : gin IX
      2      2      3
o55 = monomialIdeal (x , x*y , y )

o55 : MonomialIdeal of QQ [x, y, z]
```

A word of caution is needed here. The program works by making a random change of variables by picking a random matrix $g \in GL_n(k)$. However, every once in a while, a “bad” choice of g will occur because we may pick a g that is outside the Zariski open set of Lemma 4.1. To observe this behavior, try running the command

```
i56 : apply(1..100, j-> print gin IX)
```

If you apply this command enough times, eventually a wrong answer appears.

We won’t say anymore about generic initial ideals here, but the next step is to show that the generic initial ideal is an example of a Borel-fixed ideal. Since the ideal is Borel-fixed, one can then use the Eliahou-Kervaire resolution to describe the minimal free resolution of the generic initial ideal. See [MS] for all the details. A good reference on generic initial ideals is M. Green’s introductory survey [G].

SUGGESTED READINGS

The following books provide a good basis for learning more about today’s material.

- [BH] Winfried Bruns and Jürgen Herzog, *Cohen-Macaulay rings*, Cambridge Studies in Advanced Mathematics, vol. 39, Cambridge University Press, Cambridge, 1993.
- [G] Mark Green, *Generic initial ideals*, Six lectures on commutative algebra (Bellaterra, 1996), Progr. Math., vol. 166, Birkhäuser, Basel, 1998, pp. 119–186.
- [MS] Ezra Miller and Bernd Sturmfels, *Combinatorial commutative algebra*, Graduate Texts in Mathematics, vol. 227, Springer-Verlag, New York, 2005.
- [S] Richard P. Stanley, *Combinatorics and commutative algebra. Second edition*, Progress in Mathematics, vol. 41, Birkhäuser Boston, Inc., Boston, MA, 1996.
- [V] Rafael H. Villarreal, *Monomial algebras*, Monographs and Textbooks in Pure and Applied Mathematics, vol. 238, Marcel Dekker Inc., New York, 2001.